

# ALGORITMOS E PROGRAMAÇÃO DE COMPUTADORES

## AULA 4

### Estruturas de Controle de Repetição

Vandor Roberto Vilardi Rissoli



# APRESENTAÇÃO

- Estrutura de Controle de Repetição
  - para...faca
  - enquanto...faca
  - faca...enquanto
- Fluxograma
- Tratamento de Texto
- Referências



# Estrutura de Repetição

Crie um algoritmo que leia um número de 1 à 10 e mostre o cálculo da tabuada do valor lido. O resultado deve ser apresentado como o exemplo abaixo:

Exemplo:

$$2 \times 1 = 2$$

$$2 \times 2 = 4$$

$$2 \times 3 = 6$$

$$2 \times 4 = 8$$

$$2 \times 5 = 10$$

$$2 \times 6 = 12$$

$$2 \times 7 = 14$$

$$2 \times 8 = 16$$

$$2 \times 9 = 18$$

$$2 \times 10 = 20$$

Solução →

**algoritmo** tabuada;

// Síntese

// Objetivo: tabuada de um número

// Entrada: valor da tabuada

// Saída: valor calculado de 1 a 10

**principal**

// Declarações

inteiro numero;

// Instruções

escreva("Informe o número desejado: ");

leia(numero);

escreval(numero, " x 1 = ", numero\*1);

escreval(numero, " x 2 = ", numero \*2);

: : :

escreval(numero, " x 10 = ", numero \*10);

**fimPrincipal**

# Estrutura de Repetição

## PARA... FAÇA

- Executar um conjunto de ações por um número definido de vezes a partir da definição de limites fixos controlados pela variável de controle **v**.

- Sintaxe Geral:

**para** ( **<v>** de **<vi>** ate **<vf>** passo **<p>** ) **faca**

comando 1;

comando 2;

...

comando *n*;


**fimPara**

- **v** é a variável de controle
- **vi** é o valor inicial de **v**
- **vf** é o valor final de **v**
- **p** é o valor do incremento dado à variável **v**



# Fluxograma

PARA... FACA



# EXEMPLO

## • Algoritmo para calculo da média aritmética de 50 números

**algoritmo** media;

// Síntese

// Objetivo: calcular a média aritmética para 50 números

// Entrada: 50 números

// Saída: média aritmética dos 50 números

**principal**

// Declarações

real numero, soma;

inteiro auxiliar;

// Instruções

soma = 0;

**para** (auxiliar de 1 ate 50 passo 1) **faca**

escreva("Informe o valor numérico: ");

leia(numero);


soma = soma + numero;

**fimPara**

escreva ("Média = ", (soma / 50):5:2); // formatação de real

**fimPrincipal**

passo é obrigatório



{  
auxiliar = 1  
auxiliar <= 50  
auxiliar = auxiliar + 1

# Estrutura de Repetição

## FORMATAÇÃO DE DADOS NA SAÍDA

- Os resultados de expressões matemáticas devem possuir formatação coerente a realidade do problema, sendo obrigatório formatar resultados reais quando forem ser apresentados

**<expressão>:<tamanho em casas>:<qtde. de casas decimais>**

- Exemplo:

escreva(" 1 /3 = ", (1/3):**10:2**);

- Tamanho de **10** casas com **2** casas decimais (após a vírgula)

escreval(" 1 /3 = ", (1/3):**::3**)

- Qualquer tamanho total de casas com 3 casas decimais



# Exercícios de Fixação

1. Calcule o fatorial de um número usando a estrutura **PARA...FACA**, baseando-se na definição matemática ( $N! = 1 \times 2 \times 3 \times \dots \times N-1 \times N$  e  $0! = 1$  por definição)
2. Faça um algoritmo que lendo dois números calcule o resultado da multiplicação de um pelo outro sem utilizar a operação de multiplicação (ou seja, utilizando apenas soma)
3. Altere o algoritmo do exercício 1 (cálculo do Fatorial) utilizando a instrução **PARA...FACA** no processo inverso, ou seja, realizando as operações de multiplicação na ordem inversa a que foi implementada no exercício 1, sendo exigido o uso do **PARA...FACA** invertido.



# Estrutura de Repetição

## ENQUANTO... FAÇA

- Um conjunto de ações é executado repetidamente enquanto uma determinada condição permanece válida (verdadeira).
- Efetua um teste lógico antes de iniciar as instruções de repetição (ou de laço).
- O controle pode ser feito pelo usuário ou automaticamente por um contador (geralmente pela variável de controle).


Sintaxe do comando:

```
enquanto (<condição>) faça  
    comando 1;  
    comando 2;  
    ...  
    comando n;  
fimEnquanto
```



# Fluxograma

## ENQUANTO... FAÇA



# EXEMPLO

- **Algoritmo para calculo da média aritmética de 50 números**

**algoritmo** media;

// Síntese

// Objetivo: calcular a média aritmética para 50 números

// Entrada: 50 números

// Saída: média aritmética dos 50 números

**principal**

// Declarações

real numero, soma;

inteiro contador;

// Instruções

soma = 0;

contador = 1;

**enquanto** (contador <= 50) **faca**

    escreva("Informe o número", contador, " desejado:");

    leia(numero);

    contador = contador + 1;

    soma = soma + numero;

**fimEnquanto**

    escreva ("Média = ", (soma/50)::3);

**fimPrincipal**

# Exercícios de Fixação

4. Altere o algoritmo que apresenta o menor número, considerando agora a menor de 10 alturas de pessoas usando a instrução **enquanto...faca**.
5. Sendo  $H = 1 + 1/2 + 1/3 + \dots + 1/n$ . Prepare um algoritmo para calcular  $H$  sendo  $n$  fornecido pelo usuário usando somente a instrução **enquanto...faca**.
6. Elabore um algoritmo para calcular  $N!$  (fatorial de  $N$ ) sendo que o valor inteiro de  $N$  é fornecido pelo usuário. Sabe-se que:  $N! = 1 \times 2 \times \dots \times N-1 \times N$  e  $0! = 1$ . Use o **enquanto...faca**.



# Estrutura de Repetição

## FACA... ENQUANTO

- Repetição com teste condicional ao final;
- Um conjunto de ações é executado **enquanto** uma determinada condição for **VERDADEIRA**;
- Ao menos uma vez sempre será executado o bloco de repetição.
- Sintaxe do comando:

**faca**

comando 1;

comando 2;


⋮

comando  $n$ ;

**enquanto** (<condição> );

# Fluxograma

## FACA... ENQUANTO



# EXEMPLO

- **Algoritmo para calculo da média aritmética de 50 números**

**algoritmo** media;

// Síntese

// Objetivo: calcular a média aritmética para 50 números

// Entrada: 50 números

// Saída: média aritmética dos 50 números

**principal**

// Declarações

real numero, soma;

inteiro contador;

// Instruções

soma = 0;

contador = 1;

**faca**

    escreva("Informe o número ",contador, " desejado:");

    leia(numero);

    contador = contador + 1;

    soma = soma + numero;

**enquanto** (contador <= 50);

    escreval(" Media = ", (soma/50):5:4);

**fimPrincipal**

# Exercícios de Fixação

- 7) Elabore um algoritmo que permita a escolha das frutas especificadas abaixo, sendo elas identificadas por meio do seu valor correspondente.

1 => ABACAXI; 2 => MAÇA; 3 => PERA

Usando a instrução **faca...enquanto** identifique quais as frutas desejadas por seu usuário, lembrando que ele pode desejar somente um tipo de fruta de cada vez, porém pode comprar várias unidades da mesma fruta, mas sem informar a quantidade, selecionando várias vezes a opção do menu acima.

Porém, deve existir o cuidado na identificação correta das frutas, pois o feirante só possui estes três tipos. Sendo assim, o usuário que escolher uma fruta não existente deve ser notificado que a fruta não está disponível e a solicitação deve ser feita novamente sobre qual fruta ele deseja. Ao final, mostre o total de cada fruta adquirida.



# Exercícios de Fixação

- 8) Altere os exercícios anteriores, que são apresentados a seguir, elaborando-os na estrutura para...faca, ao invés de enquanto...faca, caso isso seja possível.
- a) Faça o algoritmo que apresenta o menor número, considerando a menor de 10 alturas.
  - b) Sendo  $H = 1 + 1/2 + 1/3 + \dots + 1/n$ , prepare uma algoritmo para calcular o resultado da série ( $H$ ), sendo  $n$  fornecido pelo usuário.



# Estrutura de Repetição

## CARACTERÍSTICAS DAS ESTRUTURAS DE REPETIÇÃO

ESTRUTURA	Condição de término	Quantidade de execuções	Condição de existência/repetição
<b>enquanto</b>	início	?	Condição verdadeira
<b>faca</b>	final	mínimo 1	Condição verdadeira
<b>para</b>	início	$(vf-vi+p) \setminus p$	$v \leq vf$ <b>ou</b> $v \geq vf$ (depende do passo)



# Estrutura de Repetição

As estruturas de repetição (**enquanto** e **faca**) podem utilizar:

- Uma variável auxiliar que funciona como **contador** de quantas vezes será executado o seu bloco de comandos
- Uma variável que seja utilizada como **resposta**, permitindo que o algoritmo seja executado até o valor da resposta ser coerente com um valor desejado (essa variável é muitas vezes chamada de *flag* ou *sentinela*).



## EXEMPLOS – enquanto...faca

# CONTADOR

**algoritmo automatico;**

## // Síntese

```
// Objetivo: ler 5 nomes
```

```
// Entrada : nomes
```

```
// Saída : confirmação da quantidade
```

// de nomes lidos

**principal**

## // Declarações

texto nome;

inteiro contador;

## //Instruções

```
contador = 0;
```

enquanto (contador < 5) faça

```
escreva(" Nome = ");
```

```
leia(nome);
```

```
contador = contador + 1;
```

fimEnquanto

```
escreval(contador," nomes lidos");
```

## fimPrincipal

## CONTROLADO PELO USUÁRIO

**algoritmo** pelo\_usuario;

## // Síntese

```
// Objetivo: ler os nomes informados
```

```
// Entrada : nomes
```

```
// Saída : confirmação da quantidade
```

// de nomes lidos

**principal**

## // Declarações

texto nome;

```
inteiro contador;
```

## // Instruções

```
escreva(" Nome = ");
```

```
leia(nome);
```

```
contador = 1;
```

```
enquanto (comparaTexto(nome,"FIM")
           != 0) faca
```

```
escreva(" Nome = ");
```

```
leia(nome);
```

```
contador = contador + 1;
```

fimEnquanto

```
escreval(contador, " nomes lidos");
```

## fimPrincipal

# Tratamento de Texto

Tratando a situação onde **NÃO** deve ser considerada a diferença de maiúsculo e minúsculo no tipo Texto.

**algoritmo** trata\_texto;

// Síntese

// Objetivo: ler nomes

// Entrada : nomes

// Saída : confirmação da

// quantidade dos

// nomes lidos

**principal**

// Declarações

texto nome;

inteiro contador;

//Instruções

escreva(" Nome = ");

leia(nome);

nome= **maiusculo**(nome);

contador = 1;

enquanto (**comparaTexto**(nome,"**FIM**") != 0) **faca**

escreva(" Nome = ");

leia(nome);

nome = **maiusculo**(nome);

contador = contador +1;

**fimEnquanto**

escreval(contador, " nomes lidos");

**fimPrincipal**

## Funções Embutidas

**comparaTexto**

= 0 → textos idênticos

!= 0 → textos diferentes

**maiusculo**

transforma em maiúsculo

# EXEMPLOS – **faca...enquanto**

## CONTADOR

**algoritmo** automatico;

// Síntese

// Objetivo: ler 5 nomes

// Entrada : nomes

// Saída : confirmação da quantidade

// dos nomes lidos

**principal**

// Declarações

texto nome;

inteiro contador;

// Instruções

contador = 0;

**faca**

escreva("Nome = ");

leia(nome);

contador = contador + 1;

**enquanto** (contador < 5);

escreval(contador, " nomes lidos");

**fimPrincipal**

## CONTROLADO PELO USUÁRIO

**algoritmo** pelo\_usuario;

// Síntese

// Objetivo: ler os nomes informados

// Entrada: nomes

// Saída: confirmação da quantidade de

// nomes lidos

**principal**

// Declarações

texto nome;

inteiro contador;

// Instruções

contador = 0;

**faca**

escreva(" Nome = ");

leia(nome);

nome = maiusculo(nome);

contador = contador + 1;

**enquanto** (**comparaTexto**(nome, "FIM")  
!= 0);

escreval(contador, " nomes lidos");

**fimPrincipal**

# EXEMPLO

**algoritmo** registraNome;

// Síntese

// Objetivo: armazenar nome obrigatório ao cadastro

// Entrada: nome

// Saída: confirma registro do nome

**principal**

// Declarações

texto nome;

// Instruções

faca

escreval("Digite seu nome");

leia(nome);

se (**tamanhoTexto**(nome) == 0) entao

escreval(" Nome inválido ");

fimSe

enquanto(**tamanhoTexto**(nome) == 0);

escreval(" Dado coletado com sucesso");

escreva(" Nome: ", **minusculo**(nome));

escreva(" Quantidade de Caracteres = ", **tamanhoTexto**(nome));

**fimPrincipal**

## Funções Embutidas

**tamanhoTexto**

quantidade de caracteres

**minusculo**

transforma em minúsculo

# Exercícios de Fixação

- 9) Um mercador quer fazer um aumento em todos os seus produtos de **10%**. Faça diferentes algoritmos (um para cada estrutura de repetição) que leia o preço de um produto e calcule e escreva o aumento de cada produto. Considere as seguintes situações:
- a) o mercador informa o número de produtos que possui antes de entrar com o preço dos produtos
  - b) o mercador informa produto a produto e você deve perguntar se tem mais algum produto a ser calculado o aumento
  - c) o preço do último produto fornecido pelo mercador é **-1**





# Exercícios de Fixação

10) Uma pesquisa sobre algumas características físicas da população de uma determinada região coletou os seguintes dados, referentes a cada habitante, para serem analisados:

- a) sexo (masculino, feminino);
- b) cor dos olhos (azuis, verdes, castanhos, pretos);
- c) cor dos cabelos (loiros, castanhos, pretos);
- d) idade em anos.

Para cada habitante, foi digitada um registro com esses dados e o último registro, que não corresponde a ninguém, conterá o valor de **idade** igual a **-1**.

Fazer um algoritmo que determine e escreva:

- a) a maior idade dos habitantes cadastrados;
- b) a porcentagem de indivíduos do sexo feminino cuja idade está entre 18 e 35 anos inclusive e que tenham olhos verdes e cabelos loiros.

→ Análise com atenção o dado que será usado como variável de controle e elabore uma solução adequada ao problema apresentado acima.

# Exercícios de Fixação

11) Fazer um programa que calcule e escreva a soma dos 30 primeiros termos da série:  $\frac{480}{10} - \frac{475}{11} + \frac{470}{12} - \frac{465}{13} + \dots$

12) Um concurso de piadas foi feito no interior de Goiás, e você foi escolhido para fazer um algoritmo que leia o nome e a nota dos 3 candidatos finalistas. Solicitando a quantidade de juízes participantes desta final, determine qual foi a piada vencedora, baseando-se no total de pontos conseguidos para cada piada. Lembre-se que a nota da avaliação indicada por cada juiz só pode variar de 0 (zero) até 100 (cem) pontos. Caso aconteça empate mostre o nome dos vencedores que empataram.



# Exercícios de Fixação

13) Fazer um algoritmo que leia 100 números inteiros ou até que seja lido um número negativo e mostre a quantidade exata do total de números lidos. Na solução deste problema não pode ser contabilizado nenhum valor inválido, assim como não pode ser superado o limite máximo de 100 números.

Este valor limite (100) deverá estar definido em uma constante e seu algoritmo deverá utilizar esta constante sempre que sua lógica necessitar trabalhar sobre a quantidade limite de valores a serem solicitados ao usuário.



# Referências de Criação para Apoio ao Estudo

## Material para Consulta e Apoio ao Conteúdo

- FARRER, H. *et al.* Algoritmos Estruturados, Editora LTC, 3ª . edição, 1999.
  - Capítulo 1
- MANZANO, J.; OLIVEIRA, J. Algoritmos, Lógica para desenvolvimento de programação, Editora Ética, 1996.
  - Capítulo 5
- Universidade de Brasília (UnB Gama)
  - <https://sae.unb.br/cae/conteudo/unbfga>  
(escolha a disciplina **Algoritmo Prog Computadores**)

